日水研報告，(34)：53—60，1984.
Bull. Jap. Sea Reg. Fish. Res. Lab., (34): 53—60, 1984.

短　報

Marqualdt 法による Polymodal な度数分布を正規分布へ分解する BASIC プログラム

赤嶺　達郎[1)]

The BASIC Program to Analyse the Polymodal Frequency Distribution into Normal Distributions with Marqualdt's Method

Tatsuro Akamine[1)]

Abstract

The algorithm of this program is Marqualdt's method. Gauss' elimination method is used to solve the simultaneous linear equations. Each parameter is scaled during calculation for faster convergence. User inputs the data and initial values of the parameters. It is adequate for convergence to set $\lambda=10000$ or larger.

Polymodal な度数分布を正規分布へ分解する手法は階級の中央値に度数をプロットすることにより，最小二乗法による回帰曲線の手法に帰着できる．赤嶺（1982）は Gauss-Seidel 法のプログラムを作成した．これは小型計算機用にメモリーを最小限に押さえたため，収束が遅いという欠点があった．

最近の小型計算機の高性能化は著しく，より収束の速いプログラムの作成が可能になってきた．このプログラムは Marqualdt 法によるもので収束も速く，安定性もよい．使用機種は PC 8801 (NEC) であるが，メモリーが十分であればそれ以下の機種でも可能である．

1) 〒951 新潟市水道町1丁目 5939—22
日本海区水産研究所
(Japan Sea Regional Fisheries Research Laboratory, Suido-cho, Niigata 951, Japan)

算　法

与えられた度数分布を F，階級の数 m，階級の幅 h，階級値の最小値 a，最大値 b とする．分解する正規分布の数を n とおくと，求める式 f，残差関数 d^2 は

$$f=\sum_{i=1}^{n}K_i\cdot N(\mu_i,\sigma_i,x) \quad \cdots\cdots\cdots\cdots ①$$

$$N(\mu_i,\sigma_i,x)=\frac{1}{\sqrt{2\pi}\sigma_i}\exp\left\{-\frac{(x-\mu_i)^2}{2{\sigma_i}^2}\right\}$$

$$d^2=\sum_{x=a}^{b}{d_x}^2=\sum_{x}(F-f)^2 \quad \cdots\cdots\cdots\cdots ②$$

$$b=a+(m-1)h$$

となる（第1図）．①より

$$\left\{\begin{aligned}\frac{\partial f}{\partial K_i}&=N(\mu_i,\sigma_i,x)\\ \frac{\partial f}{\partial \mu_i}&=K_i\cdot N(\mu_i,\sigma_i,x)\cdot\frac{x-\mu_i}{{\sigma_i}^2}\\ \frac{\partial f}{\partial \sigma_i}&=K_i\cdot N(\mu_i,\sigma_i,x)\cdot\frac{(x-\mu_i)^2-{\sigma_i}^2}{{\sigma_i}^3}\end{aligned}\right.$$

である．

1. Gauss-Seidel 法と Gauss-Newton 法

①を一次近似すると

$$\Delta f\fallingdotseq\sum_{i}^{n}\left\{\frac{\partial f}{\partial K_i}\Delta K_i+\frac{\partial f}{\partial \mu_i}\Delta\mu_i+\frac{\partial f}{\partial \sigma_i}\Delta\sigma_i\right\}$$

となる．ここでパラメータ K_i,μ_i,σ_i をまと



第1図　変数の説明

Fig. 1. Illustration of variables.

めて $\alpha_i(i=1\sim3n)$ で表わすと

$$\Delta f \fallingdotseq \sum_i^{3n} \frac{\partial f}{\partial \alpha_i} \Delta\alpha_i$$

となる．これを②に代入すると

$$d^2 = \sum_x \left\{F-(f+\Delta f)\right\}^2$$

$$=\sum_x \left(d_x - \sum_i \frac{\partial f}{\partial \alpha_i} \Delta\alpha_i\right)^2$$

ここで最小二乗法を用いると

$$\frac{\partial d^2}{\partial \alpha_j} = -2\sum_x \left(d_x - \sum_i \frac{\partial f}{\partial \alpha_i} \Delta\alpha_i\right) \frac{\partial f}{\partial \alpha_j}$$

$$=0$$

これより次の連立方程式を得る．

$$\left(\sum_x \frac{\partial f}{\partial \alpha_i}\ \frac{\partial f}{\partial \alpha_j}\right)\left(\Delta\alpha_i\right)=\left(\sum_x d_x \frac{\partial f}{\partial \alpha_i}\right) \quad \cdots③$$

これを正規方程式と呼び

$$\boldsymbol{A\Delta\alpha=b}$$

と略記する．これを解いて $\Delta\alpha_i$ を求めるのが Gauss–Newton 法であるが，実際には正規方程式を直接用いない手法の方が精度が高く大型計算機等で採用されている（中川・小柳 1982)．ここでは Gauss の消去法によって正規方程式を解く手法を採用した．ただし，係数行列は対称正定値なので，計算は上三角行列で行なっている（戸川1971).

Gauss–Seidel 法ではパラメータを1つずつ動かすので，$\Delta\alpha_j=0$ $(j\neq i)$ より③は

$$\sum_x \left(\frac{\partial f}{\partial \alpha_i}\right)^2 \Delta\alpha_i = \sum_x \left(d_x \frac{\partial f}{\partial \alpha_i}\right)$$

となり簡単に解ける．これを2変数の場合で説明すると第2図のようになり，収束が遅いことがわかる．



第2図　パラメータ数が2の場合の Gauss–Seidel 法の収束

Fig. 2. Convergence of Gauss–Seidel method in the case of 2 parameters.

これに対して Gauss–Newton 法では，回帰直線のように線型のものは1回で解が求まる．非線型のものでは線型近似による誤差のため反復が必要であるが，解の近傍では収束は非常に速い．ただし解からはなれていると振動・発散をおこす．

両手法を比較するため，田中（1956）のキダイ体長組成データの分解を行なった．Gauss–Seidel 法では完全に収束するまで500回の反復が必要であるが，Gauss–Newton 法では5回以内で完全に収束した．ただし，通常の初期値では Gauss–Newton 法で振動・発散を生じたため，Gauss–Seidel 法で5回反復計算したものを初期値として使用した（第3図).



第 3 図　d^2 の収束状態
GS : Gauss–Seidel 法　GN : Gauss–Newton 法　データ：キダイの例
初期値：Gauss–Seidel 法で 5 回反復計算

Fig. 3. Convergence of d^2.
GS: Gauss–Seidel method. GN: Gauss–Newton method. Data: the example of porgy. Initial values: after 5 interations of Gauss–Seidel method.

2. 最急降下法 (Steepest descent method)

パラメータを α_i とおくと，②より

$$\frac{\partial d^2}{\partial \alpha_i}=-2\sum_x d_x \frac{\partial f}{\partial \alpha_i} \quad \cdots\cdots ④$$

となる．ここで

$$\boldsymbol{g}=\left(-\frac{\partial d^2}{\partial \alpha_i}\right)$$

とおけば，ベクトル $\boldsymbol{g}$ は d^2 の等高面の法線であり，谷底の方向に向いている（第 4 図）．$\boldsymbol{g}$ を最急降下ベクトルと呼ぶ．そこで

$$\Delta\,\alpha_i=\left(\sum_x d_x \frac{\partial f}{\partial \alpha_i}\right)k$$

とし，k を適当にとって反復すれば解に到達できる．一般的には適当な初期値 k_0 を与え，$\Delta d^2<0$ のときには $k_{n+1}=k_n*1.2$ で加速し，$\Delta d^2\geqq 0$ のときには $k_{n+1}=k_n/2$ で減速してやり直す (Ruckdeschel 1982)．また



第 4 図　パラメータ数が 2 の場合の最急降下法の収束

Fig. 4. Convergence of steepest descent method in the case of 2 parameters.

Gauss–Seidel 法と同様の方法をとれば

$$\Delta f \fallingdotseq \sum_i \frac{\partial f}{\partial \alpha_i} \Delta \alpha_i = \left\{ \sum_i \frac{\partial f}{\partial \alpha_i} \left(\sum_x d_x \frac{\partial f}{\partial \alpha_i} \right) \right\} k$$

$$= \beta_x k$$

$$d^2 = \sum_x \left\{ F - (f + \Delta f) \right\}^2 = \sum_x (d_x - \beta_x k)^2$$

$$\frac{\partial d^2}{\partial k} = 0 \quad \text{より} \quad k = \frac{\sum_x d_x \beta_x}{\sum_x \beta_x{}^2}$$

で k が求まる.

しかし，このままでは収束しない. それは $\frac{\partial f}{\partial K_i} \ll \frac{\partial f}{\partial \mu_i}$, $\frac{\partial f}{\partial \sigma_i}$ のため，K_i が全く変化しないためである. そこで $\alpha_i' = \frac{\alpha_i}{a_i}$ でスケーリングを行なった. このとき

$$\frac{\partial f}{\partial \alpha_i'} = \frac{\partial f}{\partial \alpha_i} \frac{\partial \alpha_i}{\partial \alpha_i'} = a_i \cdot \frac{\partial f}{\partial \alpha_i}$$

$$\Delta \alpha_i = a_i \cdot \Delta \alpha_i'$$

となる. ここでは a_i としてそのときの α_i の値を使用した. したがって各パラメータのスケールはすべて1となっている（第5図).



第5図 パラメータのスケーリング
$\boldsymbol{g}$ ：最急降下ベクトル
$\boldsymbol{g}'$：スケーリングしたパラメータの最急降下ベクトル

Fig. 5. Scaling of parameters.
$\boldsymbol{g}$: steepest descent vector.
$\boldsymbol{g}'$: steepest descent vector of scaled parameter.

このスケーリングにより収束はかなりよくなる. 単純な場合にはこれでも十分であるが，この場合には収束が遅くて使い物にならない（第1表). これは最急降下方向は局所的には最良の方向であるが，大局的には必ずしもよい方向でないためである（中川・小柳 1982). このように最急降下法は，安定であるが収束は遅いという特徴をもっている.

3. Marqualdt 法

Gauss–Newton 法と最急降下法という互いに反対の性質を持つ手法を融合した手法が Marqualdt 法である. ③と④を比較すると $\boldsymbol{b} = \boldsymbol{g}$ である. そこで

$$(\boldsymbol{A} + \lambda \boldsymbol{I}) \Delta \boldsymbol{\alpha} = \boldsymbol{b}$$

$\boldsymbol{I}$：単位行列

とおく. λ が大きいときには $\lambda \boldsymbol{I} \Delta \boldsymbol{\alpha} = \boldsymbol{b}$ に近づき，$\Delta \boldsymbol{\alpha} = (1/\lambda) \boldsymbol{b}$ となって最急降下法となる. λ が小さいときには $\boldsymbol{A} \Delta \boldsymbol{\alpha} = \boldsymbol{b}$ に近づき，Gauss–Newton 法となる. したがって λ の初期値を大きくとり，だんだん小さくしていけば解に到達できる. Marqualdt 法では $\Delta d^2 < 0$ のときには $\lambda = \lambda / \nu$ で λ を小さくし，$\Delta d^2 \geqq 0$ のときには $\lambda = \lambda \cdot \nu$ で λ を大きくしてやり直す. 一般的には $\nu = 2$ である（嶋津 1979).

$$\Delta d^2 \fallingdotseq -2 \sum_x d_x \left(\sum_i \frac{\partial f}{\partial \alpha_i} \Delta \alpha_i \right)$$

で Δd^2 を1次近似にて求める方法もあるが，これだと振動・発散を生じた. これは非線型性が強く，1次近似では誤差が大きいためと考えられる. ここでは $\Delta d^2 = d^2{}_{n+1} - d^2{}_n$ で直接 Δd^2 を求めて判定に使用した.

応　用　例

PC 8801 を使用して田中（1956）のデータの分解を行なった. 結果を第1表に示す. λ を1000以上にとれば収束は非常によい. λ を100以下にとると2回目以後 $\lambda \to \infty$ となり中

第1表　d^2 の収束状態

Table 1.　Convergence of d^2.

Number of iterations	Marqualdt's method		Gauss–Seidel method	Steepest descent method
	λ=1000	$\lambda=10^6$		
0	798925	798925	798925	798925
1	30880	21717	95838	638443
2	6784	9405	25020	579662
3	6282	7646	11443	520681
4	6251	6810	8309	492315
5	6250	6530	7364	465906
6	6250	6453	6987	440723
7		6401	6802	417380
8		6359	6695	395116
9		6329	6626	374430
10		6300	6577	354749
11		6273	6539	336417
12		6257	6508	319014
13		6251	6481	302771
14		6250	6458	287377
15		6250	6439	272987
16			6421	259367
17			6406	246620
18			6392	234566
19			6380	223274
20			6370	212604

止した．これは λ が小さすぎたため，λ の寄与が小さく1回目の計算で Gauss–Newton 法によって変動し，停留点にはいったためと思われる．

Marqualdt 法では λ を大きめにとって，最急降下法に近い状態で開始しなくてはならない．この場合 λ は単調に減少して解に収束する．逆に λ が単調に増加するような場合は，解に到達したか，λ が小さすぎて停留点や他の極小点にはいったためで，中止しなくてはならない．このプログラムでは λ が連続して10回増加した場合には中止するようにしている．

λ の値としては 10000 くらいが適当なようである．PC 8801 では順調な場合1回に3分25秒程度かかった．

考　　察

一般的に d^2 は比較的急速に収束するが，各パラメータの収束は遅い．これは解の近傍ではパラメータの偏導関数の値が0に近いためと考えられ，各種の最適化法に見られる共通の特性である．したがって，パラメータの計算値に対して過度の信頼を置かない方がよい（Ruckdeschel 1982）．

最小二乗法によって回帰曲線を求める方法は，中川・小柳（1982）にくわしい．これに紹介されている **SALS** のように，大型計算機用には複雑で大きなプログラムが用意されている．しかし，小型計算機用のプログラムとしては，この程度のプログラムが適当であるように思われる．このプログラムで対応できないようなデータについては大型計算機で扱

プログラムリスト（キダイのデータの入力例）

Program list (DATA: the example of porgy)

```
10   REM
20   REM POLYMODAL 3
30   REM
40   REM ﾆｭｳﾘｮｸ
50   READ NND,MCM,NCL,CWD,NIT,LAMBDA,NU
60   PRINT "ｾｲｷﾌﾞﾝﾌﾟ ﾉ ｶｽﾞ=";NND
70   PRINT "ｻｲｼｮｳ ﾉ ｶｲｷｭｳﾁ=";MCM
80   PRINT "ｶｲｷｭｳ ﾉ ｶｽﾞ    =";NCL
90   PRINT "ｶｲｷｭｳ ﾉ ﾊﾊﾞ    =";CWD
100  PRINT "     ﾊﾝﾌﾟｸｽｳ    =";NIT
110  PRINT "     LAMBDA     =";LAMBDA
120  PRINT "       NU       =";NU
130  PRINT
140  STOP
150  N3=3*NND
160  DIM F(NCL),X(NCL),DX(NCL),BIBUN(N3),ND(NND,NCL)
170  DIM HENSU(N3),KEISU(N3,N3),TEISU(N3),ZOBUN(N3),HENSU2(N3)
180  FOR K=1 TO NCL
190     X(K)=MCM+(K-1)*CWD
200     READ F(K)
210     PRINT "F(";X(K);")=";F(K)
220  NEXT K
230  PRINT
240  STOP
250  FOR I=1 TO NND
260     S3=I:S2=S3+NND:S1=S2+NND
270     READ HENSU(S1),HENSU(S2),HENSU(S3)
280     PRINT "I=";I
290     PRINT "ﾒﾝｾｷ=";HENSU(S1),"ﾍｲｷﾝ=";HENSU(S2),"ﾌﾞﾝｻﾝ=";HENSU(S3)
300  NEXT I
310  STOP
500  REM ｹｲｻﾝ
510  P9=.398942:P8=-.5
520  FOR I=1 TO N3
530     HENSU2(I)=HENSU(I)
540  NEXT I
550  GOSUB *CLD2
560  D2=D3
570  PRINT "D2=";D2
580  PRINT
590  FOR KK1=1 TO NIT+1
600     REM ｼｮｷｶ
610     FOR I=1 TO N3
620        TEISU(I)=0
630        FOR J=I TO N3
640           KEISU(I,J)=0
650     NEXT J:NEXT I
660     REM ｿｳﾜ
670     FOR K=1 TO NCL
680        FOR I=1 TO NND
690           S3=I:S2=S3+NND:S1=S2+NND
700           P1=HENSU(S1):P2=X(K)-HENSU(S2):P3=HENSU(S3)
710           P6=ND(I,K)
720           BIBUN(S1)=P6/P1*HENSU(S1)
730           BIBUN(S2)=P6*P2/P3/P3*HENSU(S2)
740           BIBUN(S3)=P6*(P2*P2-P3*P3)/P3/P3/P3*HENSU(S3)
750        NEXT I
760        D1=DX(K)
```

```
770        REM ｹｲｽｳ
780        FOR I=1 TO N3
790           TEISU(I)=TEISU(I)+D1*BIBUN(I)
800           FOR J=I TO N3
810              KEISU(I,J)=KEISU(I,J)+BIBUN(I)*BIBUN(J)
820     NEXT J:NEXT I:NEXT K
830     LAMBDA2=0
840     K2=0
850    *REP
860     K2=K2+1
870     IF K2>11 GOTO *OWARI
880     PRINT
890     PRINT "LAMBDA=";LAMBDA
900     PRINT
910     FOR I=1 TO N3
920        KEISU(I,I)=KEISU(I,I)+LAMBDA-LAMBDA2
930     NEXT I
940     REM ｾﾞﾝｼﾝ ｼｮｳｷｮ
950     FOR I=1 TO N3-1
960        FOR K=I+1 TO N3
970           Q1=KEISU(I,K)/KEISU(I,I)
980           TEISU(K)=TEISU(K)-Q1*TEISU(I)
990           FOR J=K TO N3
1000              KEISU(K,J)=KEISU(K,J)-Q1*KEISU(I,J)
1010    NEXT J:NEXT K:NEXT I
1020    REM ｺｳﾀｲ ﾀﾞｲﾆｭｳ
1030    ZOBUN(N3)=TEISU(N3)/KEISU(N3,N3)
1040    FOR I=N3-1 TO 1 STEP -1
1050       T1=TEISU(I)
1060       FOR J=I+1 TO N3
1070          T1=T1-ZOBUN(J)*KEISU(I,J)
1080       NEXT J
1090       ZOBUN(I)=T1/KEISU(I,I)
1100    NEXT I
1110    FOR I=1 TO N3
1120       ZOBUN(I)=ZOBUN(I)*HENSU(I)
1130       HENSU2(I)=HENSU(I)+ZOBUN(I)
1140    NEXT I
1150    GOSUB *CLD2
1160    IF D3>=D2 THEN LAMBDA2=LAMBDA:LAMBDA=LAMBDA*NU:GOTO *REP
1170    REM ｼｭｳｾｲ
1180    LAMBDA=LAMBDA/NU
1190    D2=D3
1200    FOR I=1 TO N3
1210       HENSU(I)=HENSU2(I)
1220    NEXT I
1230    GOSUB *SHUTU
1240 NEXT KK1
1250 *OWARI
1260  END
1500 *CLD2
1510  D3=0
1520  FOR K=1 TO NCL
1530     F1=0
1540     FOR I=1 TO NND
1550        S3=I:S2=S3+NND:S1=S2+NND
1560        P1=HENSU2(S1):P2=X(K)-HENSU2(S2):P3=HENSU2(S3)
1570        ND(I,K)=P9*P1/P3*EXP(P8*P2*P2/P3/P3)
1580        F1=F1+ND(I,K)
1590     NEXT I
1600     D1=F(K)-F1
1610     DX(K)=D1
1620     D3=D3+D1*D1
1630  NEXT K
1640 RETURN
2000 *SHUTU
```

```
2010  REM ｼｭﾂﾘｮｸ
2020  PRINT "D2=";D2
2030  PRINT
2040  PRINT "ﾊﾝﾌﾟｸｽｳ=";KK1
2050  FOR I=1 TO NND
2060     PRINT "I=";I
2070     S3=I:S2=S3+NND:S1=S2+NND
2080     PRINT "ﾒﾝｾｷ=";HENSU(S1),"ﾍｲｷﾝ=";HENSU(S2),"ﾌﾞﾝｻﾝ=";HENSU(S3)
2090  NEXT I
2100 RETURN
3000 DATA 5,7.5,29,1,20,10000,2
3010 DATA 7,79,509,2240,2341,623,476,1230,1439,921,448,512,719,673
3020 DATA 445,341,310,228,168,140,114,64,22,0,2,2,0,0,1
3030 DATA 5000,11,1,4000,15.5,1,3000,20,1.5,1000,24,1.5,500,27,1.5
```

変数の説明

Correspondence of variables

NND : number of normal distributions

MCM : minimum class mark

NCL : number of classes

CWD : class width

NIT : number of iterations

F(K) : F

X(K) : x

DX(K) : d_x

D2 : d_n^2

D3 : d_{n+1}^2

HENSU(I) : $\alpha_{i,n}$

HENSU 2 (I) : $\alpha_{i,n+1}$

BIBUN (I) : $\dfrac{\partial f}{\partial \alpha_i}$

KEISU (I, J) : $\sum_x \dfrac{\partial f}{\partial \alpha_i} \dfrac{\partial f}{\partial \alpha_j}$

TEISU (I) : $\sum_x d_x \dfrac{\partial f}{\partial \alpha_i}$

ZOBUN (I) : $\varDelta\alpha_i$

ND(J, K) : $K_i \cdot N(\mu_i, \sigma_i, x)$

った方がよい．また，嶋津（1979）のように残差関数を積分で定義する方法も小型計算機では時間がかかりすぎるように思われる．

このプログラムは一般の回帰曲線に応用できる．偏微分を差分近似で求めて，汎用プログラムにすることも可能であるが，誤差による収束の不安定化等を考えると，そのつど作り直した方がよいと思われる．

本論を終えるにあたり，助言していただいた南西海区水産研究所内海資源部石岡清英主任研究官，日本海区水産研究所資源部加藤史彦主任研究官に深謝します．

文　献

赤嶺達郎（1982）．Polymodal な度数分布を正規分布へ分解する BASIC プログラム．日水研報告(33)：163-166.

中川　徹・小柳義夫（1982）．最小二乗法による実験データ解析―プログラム **SALS**．東京大学出版会，東京，206 pp.

Ruckdeschel, F. R.（1982）．科学計算のための BASIC サブルーチン集 2（下）（長谷川勝也，石原辰雄訳）．現代数学社，東京，245 pp.

嶋津靖彦（1979）．体長組成から年齢組成を推定する一方法．昭和 54 年度漁業資源研究会議，西日本底魚部会会議報告：36-48.

田中昌一（1956）．Polymodal な度数分布の一つの取扱方及びそのキダイ体長組成解析への応用．東海水研報告(14)：1-13.

戸川隼人（1971）．マトリクスの数値計算．オーム社，東京，323 pp.